# Contributor: Alexander Georgievskiy <galeksandrp@gmail.com>

_realname=cccl
pkgbase=mingw-w64-${_realname}
pkgname="${MINGW_PACKAGE_PREFIX}-${_realname}"
pkgver=1.0
pkgrel=1
pkgdesc="Unix cc compiler to Microsoft's cl compiler wrapper (mingw-w64)"
arch=('any')
license=('GPL')
url="https://github.com/swig/${_realname}"
source=("${url}/archive/${_realname}-${pkgver}.tar.gz")
sha256sums=('aeb456d36dc5c824b4db334286f24dbedddd026d600d24f58e62a60a2d2ff901')

prepare() {
  cd ${srcdir}/${_realname}-${_realname}-${pkgver}

  ./bootstrap
}

build() {
  cd ${srcdir}/${_realname}-${_realname}-${pkgver}

  ./configure --prefix=${MINGW_PREFIX}
  make
}

package() {
  cd ${srcdir}/${_realname}-${_realname}-${pkgver}

  make DESTDIR=${pkgdir} install
}
